// File: Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "./Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: SafeMath.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (utils/math/SafeMath.sol)

pragma solidity ^0.8.0;

// CAUTION
// This version of SafeMath should only be used with Solidity 0.8 or later,
// because it relies on the compiler's built in overflow checks.

/**
 * @dev Wrappers over Solidity's arithmetic operations.
 *
 * NOTE: `SafeMath` is generally not needed starting with Solidity 0.8, since the compiler
 * now has built in overflow checking.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryAdd(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            uint256 c = a + b;
            if (c < a) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function trySub(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b > a) return (false, 0);
            return (true, a - b);
        }
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryMul(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
            // benefit is lost if 'b' is also tested.
            // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
            if (a == 0) return (true, 0);
            uint256 c = a * b;
            if (c / a != b) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the division of two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryDiv(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a / b);
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryMod(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a % b);
        }
    }

    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        return a + b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return a - b;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        return a * b;
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator.
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return a / b;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return a % b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {trySub}.
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b <= a, errorMessage);
            return a - b;
        }
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a / b;
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting with custom message when dividing by zero.
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {tryMod}.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a % b;
        }
    }
}

// File: staking.sol



// SPDX-License-Identifier:MIT
pragma solidity 0.8.18;

import "./SafeMath.sol";
import "./IERC20.sol";
import "./Ownable.sol";

contract Staking is Ownable {
  using SafeMath for uint256;

  struct Staker {
    address user;
    uint256 amount;
    STATUS status;
    uint256 startDay;
  }

  enum STATUS {
    ACTIVE,
    COMPLETED
  }

  uint256 startStakingDate;
  uint256 amountPerDay;

  Staker[] internal stakers;

  bool startStaking = false;
  uint256 dayInSeconds = 24 * 60 * 60;

  IERC20 token;
  IERC20 xCrediToken;

  uint256 periodInDays = 181;
  uint256 daysForStaking = periodInDays * dayInSeconds;
  uint256 minimumStakingAmount = 0;

  mapping(uint256 => uint256) increase;
  mapping(address => uint256[]) stakingsMap;

  event Received(address, uint256);
  event Staked(address indexed user, uint256 amount, uint256 startDay);
  event Claimed(address indexed user, uint256 amount, uint256 date);

  constructor(address _token, address _xCrediToken) {
    require(_token != address(0), "Invalid address for token");
    require(_xCrediToken != address(0), "Invalid address for xCrediToken");

    token = IERC20(_token);
    xCrediToken = IERC20(_xCrediToken);
  }

  modifier calculation(uint256 index) {
    require(stakers.length > index, "Incorrect staker");
    require(stakers[index].user == address(msg.sender), "Not owner");
    _;
  }

  function params()
    external
    view
    returns (
      address,
      address,
      uint256,
      uint256
    )
  {
    return (address(token), address(xCrediToken), periodInDays, minimumStakingAmount);
  }

  receive() external payable {
    emit Received(msg.sender, msg.value);
  }

  fallback() external payable {
    emit Received(msg.sender, msg.value);
  }

  function getCurrentDate() external view returns (uint256) {
    return block.timestamp;
  }

  function start() external onlyOwner {
    require(startStaking == false, "Staking started");
    require(xCrediToken.balanceOf(address(this)) > 0, "Token balance: 0");

    startStaking = true;
    startStakingDate = block.timestamp;
    amountPerDay = xCrediToken.balanceOf(address(this)).div(periodInDays);
  }

  function withdraw() external onlyOwner {
    uint256 balance = address(this).balance;
    require(balance > 0, "Insufficient balance");
    address payable owner = payable(owner());
    owner.transfer(balance);
  }

  function withdrawToken(IERC20 _token) external onlyOwner {
    uint256 balance = _token.balanceOf(address(this));
    require(balance > 0, "Insufficient balance");
    require(_token.transfer(address(owner()), balance), "Transfer failed");
  }

  function getBalance() external view returns (uint256) {
    return address(this).balance;
  }

  function getAmountPerDay() external view returns (uint256) {
    return amountPerDay;
  }

  function getEndDate() external view returns (uint256) {
    return _getEndDate();
  }

  function _getEndDate() internal view returns (uint256) {
    return (startStakingDate.add(daysForStaking));
  }

  function getStartStakingDate() external view returns (uint256) {
    return startStakingDate;
  }

  function getTokenBalance() external view returns (uint256) {
    return token.balanceOf(address(this));
  }

  function getXCrediBalance() external view returns (uint256) {
    return xCrediToken.balanceOf(address(this));
  }

  function getStakers() external view returns (Staker[] memory) {
    return stakers;
  }

  function stake(uint256 amount) external {
    _stake(amount);
  }

  function weight(uint256 index) external view returns (uint256) {
    require(index < periodInDays, "Period should be lower");
    uint256 w = 0;

    for (uint256 ti = 0; ti < index; ti++) {
      w = w.add(increase[ti]);
    }

    return w;
  }

  function getCurrentDay() external view returns (uint256) {
    uint256 startDate = (block.timestamp.sub(startStakingDate)).div(dayInSeconds);
    return startDate;
  }

  function _stake(uint256 amount) internal {
    require(startStaking == true, "Staking not started");
    require(amount >= minimumStakingAmount, "Amount should be greater than 1000");

    require(_getEndDate() > block.timestamp, "Staking period finished");
    require(amount > 0, "Amount should be greater than 0");
    require(amount <= token.balanceOf(address(msg.sender)), "Insufficient USDC balance");
    require(token.transferFrom(address(msg.sender), address(this), amount), "Transfer failed");

    uint256 startDate = (block.timestamp.sub(startStakingDate)).div(dayInSeconds);

    stakers.push(Staker(address(msg.sender), amount, STATUS.ACTIVE, startDate));
    stakingsMap[address(msg.sender)].push(stakers.length - 1);

    increase[startDate] = increase[startDate].add(amount);

    emit Staked(address(msg.sender), amount, startDate);
  }

  function userStakings() external view returns (uint256[] memory) {
    return stakingsMap[address(msg.sender)];
  }

  function userStaking(uint256 index) external view returns (Staker memory) {
    return stakers[index];
  }

  function calculateRewards(uint256 index) external view returns (uint256, bool) {
    return _calculateRewards(index);
  }

  function _calculateRewards(uint256 index) internal view calculation(index) returns (uint256, bool) {
    uint256 rewards = 0;
    uint256 w = 0;
    bool calimable = false;
    (uint256 currentDay, uint256 startDay) = getDays(index);

    for (uint256 ti = 0; ti < periodInDays; ti++) {
      w = w.add(increase[ti]);

      if (currentDay > ti && startDay <= ti) {
        if (w > 0) {
          uint256 amount = stakers[index].amount.mul(amountPerDay).div(w);
          rewards = rewards.add(amount);
        }
      }

      if (ti >= currentDay) {
        break;
      }
    }

    if (currentDay > periodInDays.sub(1)) {
      calimable = true;
    }

    return (rewards, calimable);
  }

  function getDays(uint256 index) internal view returns (uint256, uint256) {
    uint256 currentDay = (block.timestamp.sub(startStakingDate)).div(dayInSeconds);
    uint256 startDay = stakers[index].startDay;

    return (currentDay, startDay);
  }

  function withdrawPrincipal(uint256 index, uint256 requestAmount) external calculation(index) {
    require(stakers[index].status == STATUS.ACTIVE, "Staking is completed");
    require(stakers[index].amount >= requestAmount, "Request amount should be lower than staked amount");

    (uint256 currentDay, uint256 startDay) = getDays(index);

    require(currentDay < periodInDays, "Staking period is over, please request a reward claim");

    if (requestAmount < stakers[index].amount) {
      require(stakers[index].amount.sub(requestAmount) >= minimumStakingAmount, "Rest USDC amount should be greater than 1000");
      uint256 restAmount = stakers[index].amount.sub(requestAmount);
      stakers[index].amount = restAmount;
    } else {
      stakers[index].status = STATUS.COMPLETED;
      stakers[index].amount = 0;
    }
    increase[startDay] = increase[startDay].sub(requestAmount);
    require(token.transfer(stakers[index].user, requestAmount), "Transfer failed");
  }

  function claim(uint256 index) external calculation(index) {
    require(stakers[index].status == STATUS.ACTIVE, "Staking is completed");
    (uint256 currentDay, uint256 startDay) = getDays(index);

    require(currentDay > periodInDays.sub(1), "Staking is not over");

    (uint256 rewards, bool claimable) = _calculateRewards(index);

    Staker memory staker = stakers[index];
    uint256 amount = staker.amount;

    require(token.transfer(staker.user, amount), "Transfer failed USDC");
    require(xCrediToken.transfer(staker.user, rewards), "Transfer failed xCREDI");

    stakers[index].status = STATUS.COMPLETED;

    emit Claimed(staker.user, rewards, block.timestamp);
  }

}